Inflection circle

Hans-Peter Schröcker
July 2010

Consider one-parametric motion of a moving frame with respect to a fixed frame in the Euclidean plane. A point x in the moving frame has the trajectory $x(t)$ in the fixed frame.

This worksheet demonstrates that the locus of points x in the moving frame whose trajectory $x(t)$ has an inflection point at a specific time t is a circle – the *inflection circle.*

```
> restart: with(LinearAlgebra):
```

Define the motion, first rotation matrix, then translation vector:

```
> A(t) := Matrix([[cos(phi(t)), -sin(phi(t))],
                  [sin(phi(t)),  cos(phi(t))]]);
```

$$A(t) := \begin{bmatrix} \cos(\phi(t)) & -\sin(\phi(t)) \\ \sin(\phi(t)) & \cos(\phi(t)) \end{bmatrix} \tag{1}$$

```
> a(t) := Vector([a1(t), a2(t)]);
```

$$a(t) := \begin{bmatrix} a1(t) \\ a2(t) \end{bmatrix} \tag{2}$$

Trajectory of $x = (x_1, x_2)$:

```
> x(t) := A(t) . Vector([x1, x2]) + a(t);
```

$$x(t) := \begin{bmatrix} \cos(\phi(t))\, x1 - \sin(\phi(t))\, x2 + a1(t) \\ \sin(\phi(t))\, x1 + \cos(\phi(t))\, x2 + a2(t) \end{bmatrix} \tag{3}$$

First and second derivative vectors with respect to time t:

```
> dx(t) := map(diff, x(t), t);
```

$$dx(t) := \begin{bmatrix} -\sin(\phi(t)) \left(\frac{d}{dt}\phi(t)\right) x1 - \cos(\phi(t)) \left(\frac{d}{dt}\phi(t)\right) x2 + \frac{d}{dt} a1(t) \\ \cos(\phi(t)) \left(\frac{d}{dt}\phi(t)\right) x1 - \sin(\phi(t)) \left(\frac{d}{dt}\phi(t)\right) x2 + \frac{d}{dt} a2(t) \end{bmatrix} \tag{4}$$

```
> ddx(t) := map(diff, dx(t), t);
```

$$ddx(t) := \Bigg[\Bigg[-\cos(\phi(t)) \left(\frac{d}{dt}\phi(t)\right)^2 x1 - \sin(\phi(t)) \left(\frac{d^2}{dt^2}\phi(t)\right) x1 \tag{5}$$

$$+ \sin(\phi(t)) \left(\frac{d}{dt}\phi(t)\right)^2 x2 - \cos(\phi(t)) \left(\frac{d^2}{dt^2}\phi(t)\right) x2 + \frac{d^2}{dt^2} a1(t)\Bigg],$$

$$\Bigg[-\sin(\phi(t)) \left(\frac{d}{dt}\phi(t)\right)^2 x1 + \cos(\phi(t)) \left(\frac{d^2}{dt^2}\phi(t)\right) x1 - \cos(\phi(t)) \left(\frac{d}{dt}\phi(t)\right)^2 x2$$

$$-\sin(\phi(t))\left(\frac{d^2}{dt^2}\phi(t)\right)x2+\frac{d^2}{dt^2}a2(t)\Big]\Big]$$

The condition for an inflection point is linear dependence of the first and second derivative vector:

```
> w := Determinant(Matrix([dx(t), ddx(t)])):
```

We do not display this lengthy expression. A simply analysis of its degree in x_1 and x_2 and the coefficients to x_1^2, x_2^2 is enough to show that the sought locus is a circle:

```
> degree(w, {x1, x2});
```

$$2 \tag{6}$$

```
> simplify(coeff(w, x1, 2)); simplify(coeff(w, x2, 2));
```

$$\left(\frac{d}{dt}\phi(t)\right)^3$$

$$\left(\frac{d}{dt}\phi(t)\right)^3 \tag{7}$$

```
> coeff(coeff(w, x1, 1), x2, 1);
```

$$0 \tag{8}$$